NAG Fortran Library Routine Document

F07HNF (ZPBSV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07HNF (ZPBSV) computes the solution to a complex system of linear equations

$$AX = B,$$

where A is an n by n Hermitian positive-definite band matrix of bandwidth $(2k_d + 1)$ and X and B are n by r matrices.

2 Specification

```
SUBROUTINE F07HNF (UPLO, N, KD, NRHS, AB, LDAB, B, LDB, INFO)
INTEGER             N, KD, NRHS, LDAB, LDB, INFO
complex*16          AB(LDAB,*), B(LDB,*)
CHARACTER*1         UPLO
```

The routine may be called by its LAPACK name ***zpbsv***.

3 Description

The Cholesky decomposition is used to factor A as $A = U^H U$, if UPLO = 'U' or $A = LL^H$, if UPLO = 'L', where U is an upper triangular band matrix, and L is a lower triangular band matrix, with the same number of super-diagonals or sub-diagonals as A. The factored form of A is then used to solve the system of equations $AX = B$.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: if UPLO = 'U', the upper triangle of A is stored.

If UPLO = 'L', the lower triangle of A is stored.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the number of linear equations, i.e., the order of the matrix A.

Constraint: N ≥ 0.

3: KD – INTEGER *Input*

On entry: k_d, the number of super-diagonals of the matrix A if UPLO = 'U', or the number of sub-diagonals if UPLO = 'L'.

Constraint: KD ≥ 0.

4: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: NRHS ≥ 0.

5: AB(LDAB,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the upper or lower triangle of the Hermitian band matrix A, stored in the first KD $+ 1$ rows of the array. The jth column of A is stored in the jth column of the array AB as follows:

if UPLO = 'U', $\mathrm{AB}(k_d + 1 + i - j, j) = a_{ij}$ for $\max(1, j - k_d) \leq i \leq j$;
if UPLO = 'L', $\mathrm{AB}(1 + i - j, j) = a_{ij}$ for $j \leq i \leq \min(n, j + k_d)$.

On exit: if INFO $= 0$, the triangular factor U or L from the Cholesky factorization $A = U^H U$ or $A = LL^H$ of the band matrix A, in the same storage format as A.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HNF (ZPBSV) is called.

Constraint: LDAB $\geq$ KD $+ 1$.

7: B(LDB,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$. To solve the equations $Ax = b$, where b is a single right-hand side, B may be supplied as a one-dimensional array with length $\mathrm{LDB} = \max(1, \mathrm{N})$.

On entry: the n by r right-hand side matrix B.

On exit: if INFO $= 0$, the n by r solution matrix X.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07HNF (ZPBSV) is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

9: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the leading minor of order i of A is not positive-definite, so the factorization could not be completed, and the solution has not been computed.

7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the ***machine precision***. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A)\frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \left\|A^{-1}\right\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

F07HPF (ZPBSVX) is a comprehensive LAPACK driver that returns forward and backward error bounds and an estimate of the condition number. Alternatively, F04CFF solves $Ax = b$ and returns a forward error bound and condition estimate. F04CFF calls F07HNF (ZPBSV) to solve the equations.

8 Further Comments

When $n \gg k$, the total number of floating point operations is approximately $4n(k+1)^2 + 16nkr$, where k is the number of super-diagonals and r is the number of right-hand sides.

The real analogue of this routine is F07HAF (DPBSV).

9 Example

To solve the equations

$$Ax = b,$$

where A is the Hermitian positive-definite band matrix

$$A = \begin{pmatrix} 9.39 & 1.08 - 1.73i & 0 & 0 \\ 1.08 + 1.73i & 1.69 & -0.04 + 0.29i & 0 \\ 0 & -0.04 - 0.29i & 2.65 & -0.33 + 2.24i \\ 0 & 0 & -0.33 - 2.24i & 2.17 \end{pmatrix}$$

and

$$b = \begin{pmatrix} -12.42 + 68.42i \\ -9.93 + 0.88i \\ -27.30 - 0.01i \\ 5.31 + 23.63i \end{pmatrix}.$$

Details of the Cholesky factorization of A are also output.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07HNF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX
      PARAMETER        (NMAX=8,KDMAX=4)
      INTEGER          LDAB
      PARAMETER        (LDAB=KDMAX+1)
      CHARACTER        UPLO
```

```
      PARAMETER        (UPLO='U')
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N
*     .. Local Arrays ..
      COMPLEX *16      AB(LDAB,NMAX), B(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         X04DFF, ZPBSV
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07HNF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN
*
*        Read the upper or lower triangular part of the band matrix A
*        from data file
*
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AB(KD+1+I-J,J),J=I,MIN(N,I+KD)),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AB(1+I-J,J),J=MAX(1,I-KD),I),I=1,N)
         END IF
*
*        Read b from data file
*
         READ (NIN,*) (B(I),I=1,N)
*
*        Solve the equations Ax = b for x
*
         CALL ZPBSV(UPLO,N,KD,1,AB,LDAB,B,N,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Solution'
            WRITE (NOUT,99999) (B(I),I=1,N)
*
*           Print details of factorization
*
            WRITE (NOUT,*)
            IFAIL = 0
            IF (UPLO.EQ.'U') THEN
               CALL X04DFF(N,N,0,KD,AB,LDAB,'Bracketed','F7.4',
     +                     'Cholesky factor U','Integer',RLABS,
     +                     'Integer',CLABS,80,0,IFAIL)
            ELSE IF (UPLO.EQ.'L') THEN
               CALL X04DFF(N,N,KD,0,AB,LDAB,'Bracketed','F7.4',
     +                     'Cholesky factor L','Integer',RLABS,
     +                     'Integer',CLABS,80,0,IFAIL)
            END IF
*
         ELSE
            WRITE (NOUT,99998) 'The leading minor of order ', INFO,
     +        ' is not positive definite'
         END IF
       ELSE
         WRITE (NOUT,*) 'NMAX and/or KDMAX too small'
       END IF
       STOP
*
99999 FORMAT ((3X,4(' (',F7.4,',',F7.4,')',:)))
99998 FORMAT (1X,A,I3,A)
      END
```

9.2 Program Data

```
F07HNF Example Program Data

   4             1                                          :Values of N and KD

 (  9.39, 0.00) (  1.08,-1.73)
                (  1.69, 0.00) ( -0.04, 0.29)
                               (  2.65, 0.00) ( -0.33, 2.24)
                                              (  2.17, 0.00) :End of matrix A

 (-12.42,68.42) ( -9.93, 0.88) (-27.30,-0.01) (  5.31,23.63) :End of vector b
```

9.3 Program Results

```
 F07HNF Example Program Results

 Solution
    (-1.0000, 8.0000) ( 2.0000,-3.0000) (-4.0000,-5.0000) ( 7.0000, 6.0000)

 Cholesky factor U
                    1                  2                  3                  4
 1  ( 3.0643, 0.0000) ( 0.3524,-0.5646)
 2                    ( 1.1167, 0.0000) (-0.0358, 0.2597)
 3                                      ( 1.6066, 0.0000) (-0.2054, 1.3942)
 4                                                        ( 0.4289, 0.0000)
```